VLA SCIENTIFIC MEMORANDA NO. 132

OPTIMAL GRIDDING

Frederic R. Schwab

October, 1980

1. Introduction.

I would like to introduce criteria for the optimality[1] of the functions employed by the convolutional gridding scheme that commonly is used in producing radio interferometer maps. Earlier memoranda in this series [1,2] describe the scheme and tally a number of functions frequently used in this application. Little attention, heretofore, has been paid to the definition of objective criteria for the selection of appropriate functions, though, according to folklore, the first of the prolate spheroidal wave functions of order 0 are the "best" functions to choose. The recognition of these as optimal functions, and the definition of the criterion by which they are so, derive from a part of the well known work of Slepian and Pollak [3] and Landau and Pollak [4]. But probably better suited to our application are functions which are optimal by criteria which slightly generalize the latter. Following Rhodes, who in [5, sec.7] studies the extremal properties of the spheroidal functions (a class which includes the spheroidal wave functions; see Stratton [6]), I define a continuum of optimality criteria parametrized by a real number $\alpha > -1$. The parameter determines a weight function, and the weighting has a ready interpretation in terms of aliasing suppression: Aliasing from far-out sources, already rejected by instrumental effects, need not be as heavily suppressed as aliasing from sources closer to the region of interest. Similarly, since the radio map usually is centered upon the region of interest, one might prefer that the spurious responses occurring near the central region of the map be suppressed to a greater degree than other spurious responses.

For the choice $\alpha=0$ (no weighting), the optimal function is a 0-order prolate spheroidal wave function. For $\alpha=\pm\frac{1}{2}$, the optimal functions are simply related to periodic Mathieu functions. Solutions for integral $\alpha=n$ are related to n-th order prolate spheroidal wave functions. (The Mathieu functions are the fundamental solutions to the wave equation in a system of elliptic cylinder coordinates; the spheroidal wave functions arise from separation of the wave equation in prolate or oblate spheroidal coordinates.)

Before continuing, recall that radio interferometer observations are essentially samples of the Fourier transform (FT) of the radio brightness intensity distribution on a patch of sky, and that these samples are irregularly spaced. The patch, for present purposes, may be considered to be flat and 2-dimensional (even 1-dimensional throughout most of the development). The "gridding" operation referred to above is simply that of convolving the samples with some given function. The purpose is twofold: first, to interpolate the data onto an equi-spaced

---

1. An optimal function is not necessarily a "good" one -- rather, it is one that is best according to some well-defined (perhaps sensible) criterion.

rectangular lattice (the grid) so that a Fast Fourier Transform algorithm may be used to approximate the brightness distribution, and, second, to reduce the intensity of spurious features that are due to radio sources lying outside of the region of interest. These features, called aliases, appear because, from samples of the FT of a function f, if the samples are distributed over a finitely spaced grid, f (unless it is periodic) can only be recovered if it is confined to some bounded region and if the linear measures of the region do not exceed limits which are dependent on the grid spacing. The effect of the convolution, in addition to interpolating the measurements, is to pre-multiply the source brightness distribution by the FT, $\hat{C}$, of the function C with which the data are convolved. In this way, the assumption of bounded support may be made approximately valid. Judicious choice of C greatly suppresses the spurious features.

C is always taken to be real and symmetric about the origin, and $\hat{C}$ inherits these properties. Usually C is separable and supported in a rectangle. The computation time required for the gridding is proportional to the area of the rectangle; hence, for computational economy, the sides of the rectangle seldom measure more than 6-10 times the grid spacing.

I shall concentrate first on the definition of the optimality criteria and on the properties of the functions which are optimal, and then on the computation of the spheroidal functions. Their computation is relatively easy. In a few cases I shall compare, with these optimal functions, some of the functions that ordinarily are used in gridding.

2. Optimality Criteria.

Suppose the sky to be flat and 2-dimensional, i.e. sky=R^2, and suppose that the observer is interested only in radio emission within a region $A \subset R^2$. Let $\eta = (\eta_1, \eta_2)$ denote the sky coordinates.[2] Also suppose that C is confined to a prescribed region $B \subset R^2$; i.e. C=0 outside of B, or supp(C)=B.





A, B given.

Find optimal C.

Want $\hat{C}$ (=FT of C) to be highly concentrated in A.

It would seem that a reasonable optimality criterion might be the quantity

$$R = \frac{\iint_A |\hat{C}(\eta_1,\eta_2)|^2 d\eta_1 d\eta_2}{\int_{-\infty}^{\infty}\int_{-\infty}^{\infty} |\hat{C}(\eta_1,\eta_2)|^2 d\eta_1 d\eta_2}$$

, which measures the extent to which $\hat{C}$ is concentrated in A. The problem, then, is to find a function C, with supp(C)=B, which maximizes R. We shall also wish to include a weight function in the definition of R:

---

2. x and y are reserved for later use, when the FT will have a factor 2π in the argument of the exponential kernel.

$$R = \frac{\iint_A |\hat{C}(\eta_1,\eta_2)|^2 w(\eta_1,\eta_2)\, d\eta_1 d\eta_2}{\int_{-\infty}^{\infty}\int_{-\infty}^{\infty} |\hat{C}(\eta_1,\eta_2)|^2 w(\eta_1,\eta_2)\, d\eta_1 d\eta_2}$$

. We shall assume throughout that A and B are symmetric about the origin.

A fair body of literature, much of it reviewed and extended in Fedotowsky and Boivin [7], deals with this type of problem. Certain cases in which A and B are both rectangular, or both circular, or both ellipsoidal reduce to nice problems whose solutions satisfy well-studied second-order ordinary differential equations. Results pertaining to rectangular domains A and B, in the absence of weighting, are due to Slepian, Landau, and Pollak; those pertaining to rectangular domains with a particular separable weight function are due to Rhodes, Fedotowsky, and Boivin; and results pertaining to circular domains, in the absence of weighting, are due to Slepian.[3]

We shall restrict our attention to the case of A and B rectangular, with sides parallel to the coordinate axes. Without loss of generality, we may assume that A is the square $[-1,1]\times[-1,1]$. We shall take B to be the rectangle of linear dimensions m_1 and m_2 times the respective grid spacings (m_1 and m_2 needn't be integral). And we shall choose a separable weight function given by $w(\eta) = w_1(\eta_1)\cdot w_2(\eta_2)$, where $w_i(\eta_i) = |1-\eta_i^2|^{\alpha_i}$, with $\alpha_i > -1$ given. Sketches of the weight function look like:



The optimal solution is a separable function which is equal to $w(\eta)$ times the product of the solutions to two 1-dimensional problems of the form:

find f which maximizes $R_\alpha(c) = \dfrac{\int_{-1}^{1} |F_\alpha(c,\eta)|^2 (1-\eta^2)^\alpha\, d\eta}{\int_{-\infty}^{\infty} |F_\alpha(c,\eta)|^2 |1-\eta^2|^\alpha\, d\eta}$

where $F_\alpha(c,\eta) = \int_{-1}^{1} e^{ic\eta t}\,(1-t^2)^\alpha\, f(t)\, dt$

and where c is related to the m_i by $c_i = \pi m_i/2$.

---

3. Slepian introduced what he termed "generalized prolate spheroidal wave functions." These functions, over circular domains, have properties analogous to the functions described below. Because of their orthogonality and completeness properties, they have been proposed for use in various reconstruction algorithms in the dissertations of radio astronomy students. They are difficult to compute.

The treatment given this problem by Rhodes is summarized below. For the 2-dimensional problem, the maximal R is given by $R^{opt} = R^{opt}_{\alpha_1}(c_1) \cdot R^{opt}_{\alpha_2}(c_2)$.

The function that maximizes $R_\alpha(c)$ is among those solutions of the differential equation $(1-\eta^2)f'' - 2(\alpha+1)\eta f' + (b-c^2\eta^2)f = 0$ which remain finite at $\eta = \pm 1$. These bounded solutions of the differential equation are termed spheroidal functions. For each α and c there are countably many solutions which are distinct up to an arbitrary normalization factor and which, in the literature, are denoted by $\psi_{\alpha n}(c,\eta)$, $n=0,1,2,\ldots$. They correspond to eigenvalues $b_{\alpha n}(c)$ of the DE that may be arranged in ascending order, $0 < b_{\alpha 0}(c) < b_{\alpha 1}(c) < b_{\alpha 2}(c) < \ldots$. $\psi_{\alpha 0}(c,\eta)$, which corresponds to the smallest eigenvalue, maximizes $R_\alpha(c)$. More generally, for each α and c, the first N+1 of the spheroidal functions are the N+1 linearly independent functions of the form $F_\alpha(c,\eta)$ which are the most concentrated on $[-1,1]$, according to the criterion $R_\alpha(c)$. The latter statements may cause the reader to pause, since it is not f, but rather the truncated, weighted-kernel FT of f, $F_\alpha(c,\eta)$, whose concentration is supposed to be maximized. The statements follow because the differential equation above is equivalent to the integral equation $\nu f(\eta) = \int_{-1}^{1} e^{ic\eta t} (1-t^2)^\alpha f(t)\, dt$ $\quad (= F_\alpha(c,\eta))$, $|\eta| \le 1$. In other words, $\psi_{\alpha n}(c,\eta)$ is, apart from a multiplicative constant (the eigenvalue $\nu_{\alpha n}(c)$ of the integral equation), its own finite, weighted-kernel Fourier transform.

The spheroidal functions have a number of interesting properties. For fixed α and c, the $\psi_{\alpha n}$ are simultaneously orthogonal over two domains, both over the interval $[-1,1]$ and over the whole real line, with respect to an inner product weighted by $|1-\eta^2|^\alpha$. They are complete in the function on $[-1,1]$ square integrable with weight $|1-\eta^2|^\alpha$, and they are complete in the space of band-limited, filtered functions of the form $F_\alpha(c,\eta)$. $\psi_{\alpha n}$ is real for real η, has exactly n zeros in the interval $[-1,1]$, is nonzero at $\eta = \pm 1$, is even or odd depending on whether n is even or odd, and its analytic extension to complex η is entire. Certain of the spheroidal functions are named special functions. The relation to prolate spheroidal wave functions is $\psi_{mn}(c,\eta) = (1-\eta^2)^{-m/2} S_{m,m+n}(c,\eta)$, for $m=0,1,2,\ldots$. The relation to periodic Mathieu functions is $\psi_{-\frac{1}{2},n}(c,\eta) = ce_n(\cos^{-1}\eta, c^2/4)$ and $\psi_{\frac{1}{2},n}(c,\eta) = (1-\eta^2)^{-\frac{1}{2}} se_{n+1}(\cos^{-1}\eta, c^2/4)$.

3. Computational Recipe.

By the criteria introduced in the preceeding section, having decided upon the use of a separable function $C(u,v) = C_1(u)\, C_2(v)$ supported in a rectangle whose sides in the directions of the coordinate axes measure m_1 and m_2 times the respective grid spacings Δu and Δv in these directions, and having chosen exponents α_1 and α_2 in the separable weight function defined above, the optimal function for

gridding is of the form $C(u,v) = |1-\eta_1^2(u)|^{\alpha_1} |1-\eta_2^2(v)|^{\alpha_2} \psi_{\alpha_1 0}(c_1, \eta_1(u)) \psi_{\alpha_2 0}(c_2, \eta_2(v))$, where $\eta_1(u) = u/m_1\Delta u$, $\eta_2(v) = v/m_2\Delta v$, and[4] $c_i = \pi m_i/2$. Thus we need to approximate $\psi_{\alpha 0}(c,\eta)$ for η in the range $[0,1]$, α perhaps in the interval $[0,2]$, and, assuming $m \lesssim 10\text{-}20$, for $c \lesssim 5\pi - 10\pi$.

I have used two means of approximating $\psi_{\alpha 0}$. Hodge, in [9], gives an efficient method to calculate the eigenvalues of the differential equation for the prolate spheroidal wave functions. I have modified his procedure to yield the eigenvalues $b_{\alpha n}(c)$ of the more general DE for the spheroidal functions. $\psi_{\alpha 0}$, then, can be computed by an expansion in fractional-order Bessel functions given by Rhodes in [5, sec.3] -- the eigenvalue determines the expansion coefficients. This expansion, convergent for all real η , is a fairly efficient means of calculating the functions for values of the parameters in the range of interest to us. Given an accurate determination of the eigenvalue, a more straightforward method, if one is only interested in $|\eta|<1$, is simply to numerically integrate the DE, starting at $\eta=0$. This method, though, is guaranteed to blow-up near $\eta=\pm1$, even if exact arithmetic, but an inexact eigenvalue, is used, since the solutions to the DE are bounded at $\eta=\pm1$ for only a countable number of eigenvalues.

Fortran subroutines implementing the two methods are shown in Figures 1 and 2. The routine in Figure 2 tabulates $\psi_{\alpha 0}(c,\eta)$ for evenly-spaced arguments $0 \le \eta \le 1$; it does so by numerically integrating the DE, and replacing the inaccurate function values in the tail-end region by function values obtained from the routine in Figure 1, which uses the Bessel function expansion. I have been able to reproduce scattered published tables of 6 and 8 significant figure accuracy; however, better programs must be fairly readily available, since these functions are useful in diverse applications. Thus I neither economize on the number of terms used in the Bessel function expansion, nor check, for extreme values of the parameters, that sufficiently many terms are summed, nor economize on the order of the matrix used in approximating the eigenvalue.

The Fourier transform of C (assuming 2π in the argument of the exponential kernel of the FT) is simply given by $\hat{C}(x,y) \propto \psi_{\alpha_1 0}(c_1, \eta_1(x)) \psi_{\alpha_2 0}(c_2, \eta_2(y))$, where $\eta_1(x) = 2x\Delta u$, $\eta_2(y) = 2y\Delta y$, and, as before, $c_i = \pi m_i/2$. The gridding scheme, as implemented in most radio interferometer data reduction programs, uses either a step function approximation or an interpolatory approximation to C. When the map is corrected to compensate for the gridding, it is more appropriate to correct by the FT of the approximation to C, rather than by a direct approximation to $\hat{C}$. The weighting corresponding to negative α is of little interest, but if it were used in conjunction with a step function approximation to C, one would need to beware of the integrable singularity at $\eta=1$, and not place a step there.

---

4. This is if R is computed by integrating over the rectangle $A = \{(x,y): |x|< 1/2\Delta u, |y|< 1/2\Delta v\}$. If the integration is over a fractionally smaller rectangle, $A = \{(f_1 x, f_2 y)$: x,y as in previous line$\}$, then choose $c_i = f_i \pi m_i/2$.

4. Discussion.

Graphs of a few of the spheroidal functions are shown in Figure 3. Figures 4-7 need explanation: These are plots of $\log_{10}|\check{C}(\eta)|$, where C is defined as in Reference [1]. That is, for a source at position η on the coordinate axis, $\check{C}(\eta)$ is the ratio of $\hat{C}(\eta)$ to $\hat{C}((\eta+1) \bmod 2 - 1)$ -- or, the ratio of the strength of an alias from position η, outside of the field of view, to the strength the source would have if it actually lay within the field at the position of its alias. Since $\hat{C}(\eta) = \psi_{\alpha 0}(c,\eta)$ has no zeros in the region $|\eta| \leq 1$, $\check{C}$, for the spheroidal functions is everywhere defined. In these graphs, sharp spikes at $\eta=3,5,7,\ldots$ are of no concern because sources at these positions have their aliases at the very edge of the radio map (and the spikes are of finite amplitude since $\hat{C}(1)\neq 0$). Figure 6 compares the weighted spheroidal functions with Gaussian-tapered sinc convoluting functions; the Gaussian-tapered sinc in the lower right-hand corner of the figure is optimal in the sense that, for a weighting exponent $\alpha=0$, I have computed the function's optimal characteristic widths (for support m=6, only). This is a rather laborious procedure which could be carried out for other parametrically defined convoluting functions. Figure 7 compares weighted spheroidal functions with the so-called Kaiser-Bessel functions, which originated as low order approximations to the prolate spheroidal wave functions of order 0; I do not know the form of the Kaiser-Bessel functions for support width $m\neq 4$.

In practice, C should be chosen so that $\hat{C}$ is not too small, since inexact arithmetic is used in the computer implementation of the algorithm. Apparently the main problem arises not from magnification of observational errors (when the spheroidal functions are used, the relative signal-to-noise ratio is close to unity -- see below) but rather from magnification by $1/\hat{C}(x,y)$ of roundoff errors that occur in the FFT computation. Let ϵ denote the characteristic unit roundoff error of the floating-point arithmetic (on a machine whose number base is β, $\epsilon = \frac{1}{2}\beta^{1-k}$ if the machine rounds, and $\epsilon = \beta^{1-k}$ if the machine truncates; for the array processor now in use, $\epsilon = 2^{-27} \approx 7.5\cdot 10^{-9}$). The relative errors in the FFT do not exceed a small multiple of ϵ. Probably $\hat{C}(x,y)$ ought to exceed ϵ by an order of magnitude or so at the corners of the map ($\hat{C}$ is much larger a small distance into the interior). Table I shows $\psi_{\alpha 0}(c,1)$ for various values of the support width m ($c=\pi m/2$) and several choices of the weighting exponent α. The upper line drawn through the table delineates a conservative upper bound on m when C is taken to be the separable product of two weighted spheroidal functions of identical parameters, assuming that the full area ($|\eta_1|, |\eta_2| \leq 1$) of the map is of interest, and allowing a safety factor of 10^3; the lower line is less conservative, allowing no safety margin at the corner pixels. Both lines were drawn assuming that $\epsilon = 2^{-27}$.

Additionally, m need not be made very large because one soon reaches a point at which aliased sources and aliased sidelobes are sufficiently well suppressed, but where the dominant annoying features of the radio map are the sidelobes of outlying sources. The convolutional gridding scheme does nothing to alleviate this problem.

^ with a k-digit mantissa,

Because the spheroidal functions are essentially their own truncated weighted-kernel FT's, the relative signal-to-noise ratio (SNR), as defined in VLA Scientific Memoranda 124 and 131 [2,10], can be expressed in a simple form when the convoluting function is a weighted spheroidal function:

$$\mathrm{SNR}(\eta) \propto \psi_{\alpha 0}(c,\eta) \Big/ \sqrt{\sum_{k=-\infty}^{\infty} \psi_{\alpha 0}^{2}(c,\eta+2k)}$$ (assuming that the convolution is done exactly, not, say, by a step function approximation). Table II shows SNR(1) for the same set of convoluting functions as in Table I (the normalization is such that SNR(0)=1). Evidently, for nonnegative α, SNR(1)$\rightarrow\sqrt{2}/2$ as m$\rightarrow\infty$. SNR is monotonic on the interval [0,1], stays near unity (for moderate m) over most of the interval, and drops abruptly to SNR(1). Because SNR is of order unity, the upper limit on the useable support width, m, is governed by the floating-point unit round-off characteristic, ϵ.

To my knowledge, the only choice of weight function in the definition of the concentration parameter, R, which leads to a well-studied class of optimal separable convoluting functions is the choice $|1-\eta^2|^{\alpha}$. However, Jarem in [11] does give a numerical construction method for a broader class of optimal functions arising from weight functions of a more general form. For our purposes the spheroidal functions appear to be "good enough."

REFERENCES

1. F. R. Schwab, "Suppression of aliasing by convolutional gridding schemes", VLA Scientific Memorandum No. 129, June 1978.

2. E. W. Greisen, "The effects of various convolving functions on aliasing and relative signal-to-noise ratios", VLA Scientific Memorandum No. 131, Dec. 1979.

3. D. Slepian and H. O. Pollak, "Prolate spheroidal wave functions, Fourier analysis and uncertainty--I", Bell Syst. Tech. J., 40 (1961), 43-63.

4. H. J. Landau and H. O. Pollak, "Prolate spheroidal wave functions, Fourier analysis and uncertainty--II", Bell Syst. Tech. J., 40 (1961), 65-84.

5. D. R. Rhodes, "On the spheroidal functions", J. Res. Nat. Bur. Standards, 74B (1970), 187-209.

6. J. R. Stratton, "Spheroidal functions", Proc. Nat. Acad. Sci. USA, 21 (1935), 51-56.

7. A. Fedotowsky and G. Boivin, "Finite Fourier self-transforms", Quart. Appl. Math., 30 (1972), 235-254.

8. D. Slepian, "Prolate spheroidal wave functions, Fourier analysis and uncertainty--IV: extensions to many dimensions; generalized prolate spheroidal wave functions", Bell Syst. Tech. J., 43 (1964), 3009-3057.

9. D. B. Hodge, "Eigenvalues and eigenfunctions of the spheroidal wave equation", J. Math. Phys., 11 (1970), 2308-2312.

10. B. G. Clark, "Gridding and signal-to-noise ratios", VLA Scientific Memorandum No. 124, April 1976.

11. J. M. Jarem, "Construction of doubly orthogonal functions", J. Math. Phys., 21 (1980), 28-31.

```
      DOUBLE PRECISION FUNCTION SPHFNO(ALPHA,C,ETA)
      IMPLICIT REAL*8 (A-H,O-Z)
      DIMENSION AA(33),BB(33),CC(33),DIAG(16),SDIAG2(16),
     X   COEF(17),RJ(50)
      DATA CL/-1.D10/,ALPHAL/-1.D20/,NDIG/15/
      IF (C.EQ.CL .AND. ALPHA.EQ.ALPHAL) GO TO 100
      CL=C
      ALPHAL=ALPHA
      IP=32
      NT=16
      CSQ=C*C
C  CALCULATE RECURSION COEFFICIENTS AND SET UP MATRIX WHOSE
C  EIGENVALUES APPROXIMATE THOSE OF THE SPHEROIDAL FUNCTION D.E.:
      ALF2=ALPHA+ALPHA
      BB(1)=CSQ/(ALF2+3.D0)
      CC(1)=(ALF2+2.D0)/(ALF2+3.D0)*CSQ
      DO 10 K=2,IP,2
         FK=DFLOAT(K)
         FK2=DFLOAT(K+K)
         FKA2=FK2+ALF2
         FKA2=FK+ALF2
         AA(K+1)=FK*(FK-1.D0)/((FK2A2-1.D0)*(FK2A2+1.D0))*CSQ
         BB(K+1)=FK*(FKA2+1.D0)+(2.D0*FK*FK+FK2*(ALF2+1.D0)+ALF2-1.D0)
     X      /((FK2A2-1.D0)*(FK2A2+3.D0))*CSQ
         CC(K+1)=(FKA2+1.D0)*(FKA2+2.D0)/((FK2A2+1.D0)*(FK2A2+3.D0))*CSQ
         N=K/2
         DIAG(N)=BB(K-1)
         IF (N.LT.IP) SDIAG2(N+1)=AA(K+1)*CC(K-1)
10       CONTINUE
C  GET SMALLEST EIGENVALUE, RLAM.  (EGRTIS CALCULATES A GIVEN NUMBER OF
C  THE SMALLEST EIGENVALUES OF A REAL, SYMMETRIC TRIDIAGONAL MATRIX.
C  ONLY THE SMALLEST EIGENVALUE IS NEEDED HERE.  THE REST OF THEM
C  APPROXIMATE THE HIGHER-ORDER, EVEN-ORDER EIGENVALUES OF THE DE.)
      CALL EGRTIS(DIAG,SDIAG2,N,1,0,IER)
      RLAM=DIAG(1)
C  GET RATIOS OF THE EXPANSION COEFFICIENTS. N IS THE NUMBER OF TERMS.:
      N=MIN0(NT,IP/2)
      COEF(N+1)=0.D0
      L=N+1
      DO 20 I=1,N
         L=L-1
         K=2*L-2
         DNM=RLAM-BB(K+1)+AA(K+3)*COEF(L+1)
         COEF(L)=0.D0
         IF (DNM.NE.0.D0) COEF(L)=-CC(K-1)/DNM
20       CONTINUE
C  NOW THE RATIO OF EACH EXPANSION COEFFICIENT TO THE NEXT LOWER ONE IS
C  KNOWN.  ARBITRARILY SET FIRST COEFFICIENT TO UNITY, AND GET THE
C  OTHERS FROM THE ABOVE RATIOS:
      COEF(1)=1.D0
      DO 30 I=2,N
30       COEF(I)=COEF(I)*COEF(I-1)
C  CALCULATE PSI(C,0):
      PI=4.D0*DATAN(1.D0)
      RTPI=DSQRT(PI)
      RT2PI=DSQRT(PI+PI)
      S00=0.D0
      DO 40 I=1,N
         L=I-1
40       S00=S00+DGAMMA(DFLOAT(L)+.5D0)/DGAMMA(DFLOAT(L+1)+ALPHA)
     X      *COEF(I)
      S00=S00/(2.D0**ALPHA*RTPI)
C  CALCULATE NU:
      RNU=RTPI*COEF(1)/(2.D0**ALPHA*DGAMMA(ALPHA+1.5D0)*S00)
100   CONTINUE
      IF (ETA.NE.0.D0) GO TO 101
      SPHFNO=1.D0
      RETURN
C  EVALUATE THE BESSEL FUNCTIONS:
101   CETA=C*DABS(ETA)
      IF (ALPHA.LT.-.5D0) GO TO 110
         A=ALPHA+.5D0
         IA=A
         NB=DFLOAT(2*N)+A+1
         CALL JAPN(CETA,A-DFLOAT(IA),NB,NDIG,RJ,IER)
         GO TO 120
110   CONTINUE
         IA=0
C  EVALUATE NEGATIVE ORDER BESSEL FUNCTION:
         CALL JAMN(CETA,1.5D0+ALPHA,2,NDIG,RJ,IER)
         RJ(1)=RJ(2)
C  EVALUATE OTHERS:
         CALL JAPN(CETA,1.5D0+ALPHA,2*(N-1),NDIG,RJ(2),IER)
120   CONTINUE
C  SUM THE SERIES:
      K=-1
      SUM=0.D0
      DO 130 I=1,N
         K=K+2
130      SUM=SUM+COEF(I)*RJ(IA+K)
C  NORMALIZE SO THAT PSI(C,0)=1.0  :
      SPHFNO=RT2PI*SUM/(CETA**(ALPHA+.5D0)*RNU)
      SPHFNO=SPHFNO/S00
      RETURN
      END
```

Note: CALL JAPN(x,a,k,m,...), for $0 \le a < 1$, evaluates to about m digits the Bessel functions $J_{a+n}(x)$, $n=0,1,\ldots,k-1$.

CALL JAMN(x,a,k,m,...), for $0 < a < 1$, evaluates $J_{a-n}(x)$.

Figure 1. Routine to calculate $\psi_{\alpha 0}(c,\eta)$ for arbitrary real η.

```
      SUBROUTINE SPHTAB(ALPHA,C,TAB,N)
C  TABULATES THE SPHEROIDAL FUNCTION OF ORDER ALPHA, INDEX 0, AND
C  PARAMETER C FOR N+1 EQUALLY SPACED ARGUMENTS IN THE RANGE
C  0.GE.ETA.LE.1.  DOES SO BY NUMERICALLY INTEGRATING THE SPHEROIDAL
C  FUNCTION DIFFERENTIAL EQUATION.  NORMALIZATION IS SUCH THAT AT
C  ETA=0 THE FUNCTION IS UNITY.  RESULTS IN THE TAIL-END REGION NEAR
C  ETA=1 ARE ALWAYS INACCURATE, SO VALUES IN THE TAIL ARE REPLACED BY
C  RESULTS FROM SPHFNO, WHICH USES A BESSEL FN. EXPANSION, UNTIL THE
C  RELATIVE AGREEMENT OF THE TWO ROUTINES IS GOOD.
      IMPLICIT REAL*8 (A-H,O-Z)
      DIMENSION Y(2),CC(24),W(2,9),TAB(1)
      EXTERNAL FCN
      COMMON/STAB/ALFP1,B,CSQ
      DATA TOL/1.D-6/,EPS/1.D-6/
      ALFP1=ALPHA+1.D0
      CSQ=C*C
      TAB(1)=1.D0
      B=SPHEIG(ALPHA,C)
      X=0.D0
      Y(1)=1.D0
      Y(2)=0.D0
      IND=1
      DO 10 K=2,N
         ETA=DFLOAT(K-1)/DFLOAT(N)
         CALL DVERK(2,FCN,X,Y,ETA,TOL,IND,CC,2,W,IER)
         TAB(K)=Y(1)
10       CONTINUE
C  CALL SPHFNO TO REPLACE TAIL-END VALUES.
      TAB(N+1)=SPHFNO(ALPHA,C,1.D0)
      DO 20 K=2,N
         L=N-K+2
         ETA=DFLOAT(L-1)/DFLOAT(N)
         YY=SPHFNO(ALPHA,C,ETA)
         REAR=DABS((YY-TAB(L))/YY)
         TAB(L)=YY
         IF (REAR.LE.EPS) GO TO 30
20       CONTINUE
30    RETURN
      END
```

```
      SUBROUTINE FCN(N,ETA,Y,YP)
      IMPLICIT REAL*8 (A-H,O-Z)
      DIMENSION Y(N),YP(N)
      COMMON/STAB/ALFP1,B,CSQ
      YP(1)=Y(2)
      YP(2)=(2.D0*ALFP1*ETA*Y(2)-(B-CSQ*ETA**2)*Y(1))
     X   /((1.D0-ETA)*(1.D0+ETA))
      RETURN
      END
```

```
      DOUBLE PRECISION FUNCTION SPHEIG(ALPHA,C)
C  COMPUTES THE SMALLEST EIGENVALUE OF THE SPHEROIDAL FUNCTION
C  DIFFERENTIAL EQUATION CORRESPONDING TO ORDER ALPHA AND PARAMETER C.
      IMPLICIT REAL*8 (A-H,O-Z)
      DIMENSION D(20),SD2(20)
      N=20
C  CONSTRUCT REAL, SYMMETRIC TRIDIAGONAL MATRIX WHOSE EIGENVALUES
C  APPROXIMATE THE EVEN-ORDER EIGENVALUES OF THE D.E.  D IS THE
C  DIAGONAL, AND SD2 HOLDS THE SQUARES OF THE SUBDIAGONAL ELEMENTS.
      CSQ=C*C
      ALF2=ALPHA+ALPHA
      BB=CSQ/(ALF2+3.D0)
      CC=(ALF2+2.D0)/(ALF2+3.D0)*CSQ
      DO 10 I=1,N
         FK=DFLOAT(2*I)
         FK2=FK+FK
         FK2A2=FK2+ALF2
         FKA2=FK+ALF2
         D(I)=BB
         IF (I.EQ.N) GO TO 10
         BB=FK*(FKA2+1.D0)+(2.D0*FK*FK+FK2*(ALF2+1.D0)+ALF2-1.D0)
     X      /((FK2A2-1.D0)*(FK2A2+3.D0))*CSQ
         AA=FK*(FK-1.D0)/((FK2A2-1.D0)*(FK2A2+1.D0))*CSQ
         SD2(I+1)=AA*CC
         CC=(FKA2+1.D0)*(FKA2+2.D0)/((FK2A2+1.D0)*(FK2A2+3.D0))*CSQ
10       CONTINUE
C  EQRT1S IS A ROUTINE TO CALCULATE THE M SMALLEST EIGENVALUES OF A
C  REAL, SYMMETRIC TRIDIAGONAL MATRIX.  HERE M=1.
      CALL EQRT1S(D,SD2,N,1,0,IER)
      SPHEIG=D(1)
      RETURN
      END
```

Note: DVERK is a Runge-Kutta code which is used here to solve the initial-value problem

$$y_1'(\eta)=y_2(\eta)$$

$$y_2'(\eta)=(1-\eta^2)^{-1}\left[2(\alpha+1)\eta\, y_2(\eta)-(b_{\alpha 0}(c)-c^2\eta^2)y_1(\eta)\right]$$

with initial conditions

$$y_1(0)=1, \quad y_2(0)=0.$$

Figure 2. Speedier routine to tabulate $\psi_{\alpha 0}(c,\eta)$, $0\le\eta\le 1$.



The spheroidal functions for $\alpha = -1/2, 0, 1/2, 1, 2, 3$, $c = 6$, $n = 0$ to 8, *with* $\lambda_{\alpha n}(c) = 1/\gamma_{\alpha n}(c)$.

Figure 3. Plots of a few of the spheroidal functions (taken from Rhodes [5]).



Figure 4. Spheroidal functions (m=6) for varied weighting function exponent, $\alpha = -\frac{1}{2}, 0, \frac{1}{2}, 1, 5/4, 3/2, 2, 5$.



Figure 4 (continued).



Figure 5. Spheroidal functions (α =0,1) for varied support sizes, m=4,6,8.



Figure 6. Comparison (for m=6) of the spheroidal functions (α =0,1) with Gaussian-tapered sinc function.



[redacted]e [redacted]Com[redacted]so.. [redacted](for [redacted]) of [redacted] the spheroidal [redacted] functions [redacted] ([redacted]=0.1) [redacted] with Kaiser-Bessel f...

TABLE I. $\Psi_{\alpha 0}(c, 1)$ $(c = \pi m/2)$

	α							
m	0	½	1	3/2	2	3	4	5
1	6.7(-1)	7.4(-1)	7.9(-1)	8.2(-1)	8.4(-1)	8.7(-1)	8.9(-1)	9.1(-1)
2	2.4(-1)	3.3(-1)	4.0(-1)	4.6(-1)	5.1(-1)	5.9(-1)	6.5(-1)	6.9(-1)
3	6.5(-2)	1.1(-1)	1.5(-1)	2.0(-1)	2.4(-1)	3.2(-1)	3.9(-1)	4.4(-1)
4	1.6(-2)	3.1(-2)	5.0(-2)	7.2(-2)	9.6(-2)	1.5(-1)	2.0(-1)	2.4(-1)
5	3.7(-3)	8.2(-3)	1.5(-2)	2.3(-2)	3.3(-2)	5.9(-2)	8.8(-2)	1.2(-1)
6	8.5(-4)	2.1(-3)	4.0(-3)	6.9(-3)	1.1(-2)	2.1(-2)	3.6(-2)	5.3(-2)
7	1.9(-4)	5.0(-4)	1.1(-3)	1.9(-3)	3.2(-3)	7.2(-3)	1.3(-2)	2.2(-2)
8	4.3(-5)	1.2(-4)	2.7(-4)	5.3(-4)	9.3(-4)	2.3(-3)	4.7(-3)	8.3(-3)
9	9.5(-6)	2.8(-5)	6.7(-5)	1.4(-4)	2.6(-4)	7.0(-4)	1.6(-3)	3.0(-3)
10	2.1(-6)	6.5(-6)	1.6(-5)	3.5(-5)	6.9(-5)	2.1(-4)	5.0(-4)	1.0(-3)
11	4.5(-7)	1.5(-6)	3.9(-6)	8.9(-6)	1.8(-5)	5.9(-5)	1.5(-4)	3.4(-4)
12	9.9(-8)	3.4(-7)	9.3(-7)	2.2(-6)	4.6(-6)	1.6(-5)	4.6(-5)	1.1(-4)
13	2.1(-8)	7.7(-8)	2.2(-7)	5.4(-7)	1.2(-6)	4.4(-6)	1.3(-5)	3.3(-5)
14	4.6(-9)	1.7(-8)	5.1(-8)	1.3(-7)	2.9(-7)	1.2(-6)	3.8(-6)	1.0(-5)

TABLE II. SNR(1) for selected weighted spheroidal convoluting functions.

				α				
m	0	½	1	3/2	2	3	4	5
1	~~0.67244~~	0.74054	0.78540	0.81698	0.84045	0.87304	0.89461	0.90992
2	0.70642	0.70695	0.70757	0.70725	0.70718	0.71083	0.72033	0.73409
3	~~0.69858~~	0.70711	0.70707	0.70706	0.70711	0.70715	0.70711	0.70724
4	0.70382	0.70658	0.70707	0.70711	0.70711	0.70711	0.70711	0.70711
5	~~0.70395~~	0.70706	0.70708	0.70710	0.70711	0.70711	0.70711	0.70711
6	0.70349	0.70694	0.70711	0.70711	0.70711	0.70711	0.70711	0.70711
7	~~0.70550~~	0.70702	0.70709	0.70711	0.70711	0.70711	0.70711	0.70711
8	0.70437	0.70706	0.70711	0.70711	0.70711	0.70711	0.70711	0.70711
9	~~0.70547~~	0.70703	0.70710	0.70711	0.70711	0.70711	0.70711	0.70711
10	0.70554	0.70709	0.70710	0.70711	0.70711	0.70711	0.70711	0.70711
11	~~0.70518~~	0.70706	0.70711	0.70711	0.70711	0.70711	0.70711	0.70711
12	0.70618	0.70708	0.70710	0.70711	0.70711	0.70711	0.70711	0.70711
13	~~0.70540~~	0.70709	0.70711	0.70711	0.70711	0.70711	0.70711	0.70711
14	0.70610	0.70708	0.70711	0.70711	0.70711	0.70711	0.70711	0.70711

Note: Crossed-out numbers are likely inaccurate -- the series was slowly convergent in these cases.